Math 3660 - Spring 2022
Mathematical Models in Economics

Steven Tschantz

2/6/22

Profit maximizing firms

Assignment 7

Instructions

Save a copy of this notebook, complete the exercises, save and submit on Brightspace your final version by start of class Tues. Mar. 15.

Nikhil Jayswal

Edit the line above that reads Enter your name here, inserting instead your name. You may also want to save to a file name that includes your name. By adding your name into the file and file name, you reduce the chance I will mix-up your submission with someone else's.

Exercise

This is a challenge exercise. Rather than being given a detailed worked example of a model, you will have to formulate the mathematical interpretation and programming for yourself.

The economic model you are asked to implement and illustrate is called Stackelberg competition, or more precisely, the price setting version of Stackelberg. You can look up a description (of the quantity setting version) in Wikipedia. However, you should not try to follow all of the details presented there, but instead create your own understanding and model following the outline presented below.

We have considered differentiated products competition model where firms compete by setting prices in Nash equilibrium, each firm deciding on its profit maximizing price given the prices of the competitors. We compared this competitive outcome to the outcome if firms cooperatively priced to maximize total profits, the collusive or monopolistic outcome. Consider now a third possible model, similar to that suggested by Stackelberg, where one firm, having a dominant position in the market, will set its price first, with the knowledge that other firms will later adjust their own prices to maximize their own profits (in Nash equilibrium among the remaining firms say) given what the first firm's price is. The other firms, the followers in this model, set prices as a function of the leader's price. The leader firm knows what the follower's price functions will be, and evaluates it's own profit as a function of the price it sets accounting for the reaction of the followers, rather than, as in the Nash equilibrium, assuming that it adjusts its price independent of competitor prices taken as fixed. This can be compared to the computation of wholesale pricing given how the retailer price and subsequent demand will be determined in the supply chain model presented earlier.

Your challenge is to implement this model, illustrating the Stackelberg competition pricing and comparing it to a Nash equilibrium outcome and a monopoly outcome. Specifically,

1) Suppose there are two firms, the leader firm and the follower firm, each with their own product. Define a demand system for the products. Take demand for each product given as a function of the two prices, calibrating the demand functions to reasonably specified conditions, say where the leader firm has a large fraction of the total demand at initial prices, with appropriate own and cross elasticities of demand.

We define the demand functions for the two firms, each as a linear function of the two prices that the firms set for their products.

In[1]:=
```
qLeader[pLeader_, pFollower_] := a1 * pLeader + b1 * pFollower + c1
qFollower[pLeader_, pFollower_] := a2 * pLeader + b2 * pFollower + c2
```

The total demand function is the sum of the two demand functions defined above.

In[3]:=
```
qTotal[pLeader_, pFollower_] := qLeader[pLeader, pFollower] + qFollower[pLeader, pFollower]
```

Since we have two linear functions with a total of 6 unknown coefficients, we need 6 conditions to calibrate and define the functions precisely. These can be obtained from the two demand quantities at initial prices, and from the 4 elasticities of demand.

Suppose that we define

In[4]:=
```
pLi = 40; pFi = 35; qLi = 100; qFi = 50;
TableForm[{{pLi, qLi}, {pFi, qFi}},
 TableHeadings → {{"Leader Firm", "Follower Firm"}, {"Price", "Quantity"}}]
```

Out[5]//TableForm=

	Price	Quantity
Leader Firm	40	100
Follower Firm	35	50

Suppose that the elasticity matrix is

In[6]:=
```
elasticities = {{-2, 1}, {1.5, -1.5}};
MatrixForm[elasticities]
```

Out[7]//MatrixForm=

$$\begin{pmatrix} -2 & 1 \\ 1.5 & -1.5 \end{pmatrix}$$

We can now write down the calibration conditions.

In[8]:=
```
calibrationconditions = {qLeader[pLi, pFi] == qLi, qFollower[pLi, pFi] == qFi,
   D[ qLeader[pLeader, pFollower], pLeader] * pLeader / qLeader[pLeader, pFollower] ==
    elasticities[[1]][[1]], D[ qLeader[pLeader, pFollower], pFollower] *
     pFollower / qLeader[pLeader, pFollower] == elasticities[[1]][[2]],
   D[ qFollower[pLeader, pFollower], pLeader] * pLeader / qFollower[pLeader, pFollower] ==
    elasticities[[2]][[1]], D[ qFollower[pLeader, pFollower], pFollower] *
     pFollower / qFollower[pLeader, pFollower] == elasticities[[2]][[2]]};
calibrationconditions = calibrationconditions /. {pLeader → pLi, pFollower → pFi};
Column[calibrationconditions]
```

Out[10]=

$$40\,a1 + 35\,b1 + c1 == 100$$
$$40\,a2 + 35\,b2 + c2 == 50$$
$$\frac{40\,a1}{40\,a1+35\,b1+c1} == -2$$
$$\frac{35\,b1}{40\,a1+35\,b1+c1} == 1$$
$$\frac{40\,a2}{40\,a2+35\,b2+c2} == 1.5$$
$$\frac{35\,b2}{40\,a2+35\,b2+c2} == -1.5$$

These equations can now be solved for the values of the coefficients.

In[11]:=
```
coefficients = FindRoot[calibrationconditions,
  {{a1, -1}, {b1, 2}, {c1, 150}, {a2, 1}, {b2, 2}, {c2, 40}}]
```

Out[11]=
```
{a1 → -5., b1 → 2.85714, c1 → 200., a2 → 1.875, b2 → -2.14286, c2 → 50.}
```

The precise form of the demand functions can be obtained now.

In[12]:=
```
qLeader[pLeader_, pFollower_] = qLeader[pLeader, pFollower] /. coefficients
```

Out[12]=
```
200. + 2.85714 pFollower - 5. pLeader
```

In[13]:=
```
qFollower[pLeader_, pFollower_] = qFollower[pLeader, pFollower] /. coefficients
```

Out[13]=
```
50. - 2.14286 pFollower + 1.875 pLeader
```

2) Assume marginal costs for the products, and compute the Nash equilibrium and monopoly prices for comparison. The marginal costs should be taken as fractions of the assumed calibration initial prices so that, with the assumed elasticities, so that Nash equilibrium and monopoly pricing will be in the same range as the initial prices.

The marginal costs for the firms are assumed to be as follows.

In[14]:=
```
mcLeader = 13; mcFollower = 10;
```

We can now write down the profit functions for both the leader and follower firms.

In[15]:=
```
profitLeader[pLeader_, pFollower_] = (pLeader - mcLeader) * qLeader[pLeader, pFollower]
```

Out[15]=
```
(200. + 2.85714 pFollower - 5. pLeader) (-13 + pLeader)
```

In[16]:=
```
profitFollower[pLeader_, pFollower_] =
  (pFollower - mcFollower) * qFollower[pLeader, pFollower]
```

Out[16]=
```
(-10 + pFollower) (50. - 2.14286 pFollower + 1.875 pLeader)
```

The equations for Nash equilibrium are

In[17]:=
```
nashEquilibriumEquations = {D[profitLeader[pLeader, pFollower], pLeader] == 0,
    D[profitFollower[pLeader, pFollower], pFollower] == 0};
Column[Simplify[nashEquilibriumEquations]]
```

Out[18]=
```
92.75 + 1. pFollower == 3.5 pLeader
1. pFollower == 16.6667 + 0.4375 pLeader
```

We can now solve the Nash equilibrium equations to get the Nash equilibrium prices.

In[19]:=
```
nashEquilibriumPrices = Solve[nashEquilibriumEquations, {pLeader, pFollower}][[1]]
```

Out[19]=
```
{pLeader -> 35.7279, pFollower -> 32.2976}
```

The profits that the firms make at these prices are

In[20]:=
```
maxProfitLeader = profitLeader[pLeader, pFollower] /. nashEquilibriumPrices
```

Out[20]=
```
2582.79
```

In[21]:=
```
maxProfitFollower = profitFollower[pLeader, pFollower] /. nashEquilibriumPrices
```

Out[21]=
```
1065.39
```

In the case of a monopoly, the leader firm supplies all the demand, and sets its own price. In that case, the demand function is a function of a single price. We assume it takes a linear form.

In[22]:=
```
monopolyDemand[pM_] := a * pM + b
```

We need two calibration conditions to determine the unknown coefficients. The first condition is given by the price elasticity of demand. The second condition is the initial demand at an initial price. We take the leader’s own elasticity as the price elasticity of demand. And the initial demand is the sum of the leader and follower demand at a price equal to the leader price.

In[23]:=
```
calibrationequations =
   {D[monopolyDemand[pM], pM] * pM / monopolyDemand[pM] == elasticities[[1]][[1]],
    monopolyDemand[pLi] == qLi + qFi};
calibrationequations = calibrationequations /. {pM → pLi};
Column[calibrationequations]
```

Out[25]=

$$\frac{40\,a}{40\,a+b} == -2$$
$$40\,a + b == 150$$

We can solve these equations for the coefficients.

In[26]:=
```
monopolyCoeffs = Solve[calibrationequations, {a, b}][[1]]
```

Out[26]=

$$\left\{a \to -\frac{15}{2},\ b \to 450\right\}$$

The precise form of the monopoly demand function is then

In[27]:=
```
monopolyDemand[pM_] = monopolyDemand[pM] /. monopolyCoeffs
```

Out[27]=

$$450 - \frac{15\,pM}{2}$$

Assuming that the marginal cost of the monopoly is the same as the marginal cost of the leader firm, the profit function for the monopoly case is

In[28]:=
```
monopolyProfitFunction[pM_] = (pM - mcLeader) * monopolyDemand[pM]
```

Out[28]=

$$\left(450 - \frac{15\,pM}{2}\right)\ (-13 + pM)$$

The first order condition for max profit is

In[29]:=
```
maxMonopolyProfitCondition = D[monopolyProfitFunction[pM], pM] == 0
```

Out[29]=

$$450 - \frac{15}{2}\ (-13 + pM) - \frac{15\,pM}{2} == 0$$

We solve this equation to get the optimal price and maximum profit.

In[30]:=
```
optimalMonopolyPrice = N[Solve[maxMonopolyProfitCondition, pM][[1]]]
```

Out[30]=

$$\{pM \to 36.5\}$$

In[31]:=
```
maxMonopolyProfit = monopolyProfitFunction[pM] /. optimalMonopolyPrice
```

Out[31]=

4141.88

The optimal price for the monopoly is similar to the prices obtained in the Nash equilibrium scenario, but the profit is higher than the combined profit in the Nash equilibrium scenario.

If the firms merge, the new profit function would be

In[32]:=
```
profitMerged[pMerged_] =
  (pMerged - mcLeader) * (qLeader[pMerged, pMerged] + qFollower[pMerged, pMerged])
```
Out[32]=
```
(250. - 2.41071 pMerged) (-13 + pMerged)
```

where we use the marginal cost of the leader firm, since it has the most supply, and the total demand is the sum of the two demand functions, with the leader and follower price being the same as new price set by the merged firm.

For maximum profit, the condition is then

In[33]:=
```
maxProfitCondition = D[profitMerged[pMerged], pMerged] == 0
```
Out[33]=
```
250. - 2.41071 (-13 + pMerged) - 2.41071 pMerged == 0
```

Solving this equation, the price that gives max profit is

In[34]:=
```
maxProfitPrice = Solve[maxProfitCondition, pMerged][[1]]
```
Out[34]=
```
{pMerged → 58.3519}
```

At this price, the max profit is

In[35]:=
```
maxProfit = profitMerged[pMerged] /. maxProfitPrice
```
Out[35]=
```
4958.33
```

This is significantly more than the sum of profits obtained in the case of Nash equilibrium. The reason is the high optimal profit price.

3) Write down the follower's profit as a function of the prices. Write down the first-order condition for the follower to maximize his own profit by setting his own price given what the price the leader sets. Solve for the optimal follower price as a function of leader price. If you take a demand function of a simple form you will be able to solve algebraically for the follower price function. Otherwise, the follower price function is defined implicitly by the first-order condition and to evaluate the derivative of this function you would need to differentiate implicitly.

The follower’s profit function is

In[36]:=
```
profitFollower[pLeader, pFollower]
```
Out[36]=
```
(-10 + pFollower) (50. - 2.14286 pFollower + 1.875 pLeader)
```

The follower firm can maximize their profit when

In[37]:=
```
maxFollowerProfitCondition = D[profitFollower[pLeader, pFollower], pFollower] == 0
```
Out[37]=
```
50. - 2.14286 (-10 + pFollower) - 2.14286 pFollower + 1.875 pLeader == 0
```

This equation can be solved to obtain the follower price as a function of the leader price.

In[38]:=
```
followerPriceFunction = Solve[maxFollowerProfitCondition, pFollower][[1]]
```
Out[38]=
```
{pFollower → -0.233333 (-71.4286 - 1.875 pLeader)}
```

In[39]:=
```
followerPriceFunction = pFollower /. followerPriceFunction;
```

In[40]:=
```
pFollower[pLeader_] = followerPriceFunction
```
Out[40]=
```
-0.233333 (-71.4286 - 1.875 pLeader)
```

4) Write down the leader's profit function as a function of own price in terms of the follower's price function. Write down the first-order condition for the leader to maximize his own profit with respect to his own profit. In general, this will depend also on the derivative of the the follower's price function, determined implicitly from the follower's first-order condition. Even in the general case, you can write down two equations, for the follower maximizing profit given leader price and for the leader maximizing profit including the follower price function derivative, that can be solved simultaneously numerically for leader and follower prices. Alternatively, for a simple demand where you can solve explicitly for the follower price function, solve for the leader profit maximizing price and determine the follower price.

The leader's profit function is

In[41]:=
```
profitLeader[pLeader, pFollower]
```
Out[41]=
```
(200. + 2.85714 pFollower - 5. pLeader) (-13 + pLeader)
```

The max profit condition is written down below. Herein, we can substitute for the follower price as a function of the leader price, obtained earlier, so as to maximize follower's profit.

In[42]:=
```
maxLeaderProfitCondition = D[profitLeader[pLeader, pFollower[pLeader]], pLeader] == 0
```
Out[42]=
```
200. - 0.666667 (-71.4286 - 1.875 pLeader) - 3.75 (-13 + pLeader) - 5. pLeader == 0
```

This equation can now be solved for the leader price.

In[43]:=
```
maxLeaderProfitPrice = Solve[maxLeaderProfitCondition, pLeader][[1]]
```
Out[43]=
```
{pLeader → 39.5159}
```

Now that the leader price is obtained, the follower price can be computed from the follower price function obtained above.

In[44]:=
```
maxFollowerProfitPrice = pFollower[pLeader] /. maxLeaderProfitPrice
```
Out[44]=
```
33.9549
```

The respective profits are

In[45]:=
```
stackelbergFollowerProfit = profitFollower[pLeader, pFollower] /. maxLeaderProfitPrice /.
  {pFollower → maxFollowerProfitPrice}
```

Out[45]=
```
1229.65
```

In[46]:=
```
stackelbergLeaderProfit = profitLeader[pLeader, pFollower] /. maxLeaderProfitPrice /.
  {pFollower → maxFollowerProfitPrice}
```

Out[46]=
```
2636.59
```

5) Compare the Stackelberg competition prices and profits to those in the Nash equilibrium and monopoly cases.

We can see that the Stackelberg competition prices are higher than the Nash equilibrium prices. The profits too are higher. Compared to the monopoly case, the Stackelberg competition profits are lower and at the same time the monopoly price is *lower* than the leader price in the Stackelberg competition.

The results are summarized in the table below.

In[47]:=
```
nashLeaderPrice = pLeader /. nashEquilibriumPrices;
nashFollowerPrice = pFollower /. nashEquilibriumPrices;
nashLeaderProfit = profitLeader[pLeader, pFollower] /. nashEquilibriumPrices;
nashFollowerProfit = profitFollower[pLeader, pFollower] /. nashEquilibriumPrices;
monopolyPrice = pM /. optimalMonopolyPrice;
monopolyProfit = maxMonopolyProfit;
stackelbergLeaderPrice = pLeader /. maxLeaderProfitPrice;
stackelbergFollowerPrice = maxFollowerProfitPrice;
TableForm[{{monopolyPrice, "", monopolyProfit, ""},
  {nashLeaderPrice, nashFollowerPrice, nashLeaderProfit, nashFollowerProfit},
  {stackelbergLeaderPrice, stackelbergFollowerPrice,
   stackelbergLeaderProfit, stackelbergFollowerProfit}},
 TableHeadings → {{"Monopoly", "Nash Equilibrium", "Stackelberg Competition"},
   {"Leader Price", "Follower Price", "Leader Profit", "Follower Profit"}}]
```

Out[55]//TableForm=

	Leader Price	Follower Price	Leader Profit	Follower Prof:
Monopoly	36.5		4141.88	
Nash Equilibrium	35.7279	32.2976	2582.79	1065.39
Stackelberg Competition	39.5159	33.9549	2636.59	1229.65

Please annotated your program, describing mathematically what you are doing. (Use cells with Format>Style>Text.)