Math 3660 - Spring 2022
Mathematical Models in Economics

Steven Tschantz

1/10/22

Profit maximizing firms

Assignment 1

Instructions

Save a copy of this notebook, complete the exercises, save and submit on Brightspace your final version by start of class Tues. Jan. 25.

Nikhil Jayswal

Edit the line above that reads Enter your name here, inserting instead your name. You may also want to save to a file name that includes your name. By adding your name into the file and file name, you reduce the chance I will mix-up your submission with someone else’s.

Exercises

1. Find the coefficients in a linear demand function assuming demand is 10000 at a price of $40 and 6000 at a price of $50.

Assume a linear demand function with some slope and intercept.

```
In[1]:= demandform[p_] = demandslope * p + demandintercept
Out[1]= demandintercept + demandslope p
```

The demand is 10000, when the price is $40, and 6000, when the price is 50$.

```
In[2]:= demandconditions = {demandform[40.] == 10000, demandform[50.] == 6000}
Out[2]= {demandintercept + 40. demandslope == 10000, demandintercept + 50. demandslope == 6000}
```

We can solve the system of linear equations to obtain the slope and intercept of the demand function.

```
In[3]:= demandcoeffs = Solve[demandconditions, {demandslope, demandintercept}][[1]]
Out[3]= {demandslope → -400., demandintercept → 26 000.}
```

The linear demand function required is thus

```
In[4]:= demand[p_] = demandform[p] /. demandcoeffs
Out[4]= 26 000. - 400. p
```

2. Solve for the profit maximizing price for this demand, assuming a constant marginal cost of $30.

We devise a cost function with an unknown fixed cost, and the given marginal cost.

```
In[5]:= marginalcost = 30.
Out[5]= 30.

In[6]:= cost[q_] = fixedcost + marginalcost * q
Out[6]= fixedcost + 30. q
```

The profit function can now be defined.

```
In[7]:= profit[p_] = p * demand[p] - cost[demand[p]]
Out[7]= -fixedcost - 30. (26 000. - 400. p) + (26 000. - 400. p) p
```

The derivative of the profit function must be zero at a critical point.

```
In[8]:= maxprofitcondition = D[profit[p], p] == 0
Out[8]= 38 000. - 800. p == 0
```

Solving the equation obtained gives us the profit maximizing price.

```
In[9]:= maxprofitsoln = Solve[maxprofitcondition, p][[1]]
Out[9]= {p → 47.5}

In[10]:= maxprofitprice = p /. maxprofitsoln
Out[10]=
47.5
```

The demand at this profit maximizing price is computed below.

```
In[11]:= demandatmaxprofit = demand[maxprofitprice]
Out[11]=
7000.
```

3. Suppose the marginal cost increases from $30 to $31. Then the optimal price to charge should also go up, though perhaps not by a full dollar. The pass-through rate is the amount the (optimal) price increases for each dollar increase in the marginal cost (or better, it is the derivative of optimal price as a function of the marginal cost parameter). Repeat the profit maximizing

calculations with the higher marginal cost and determine (approximately) the pass-through rate.

We devise a new cost function with an unknown fixed cost, and the new marginal cost.

In[12]:=
```
newmarginalcost = 31.
```
Out[12]=
```
31.
```

In[13]:=
```
newcost[q_] = fixedcost + newmarginalcost * q
```
Out[13]=
```
fixedcost + 31. q
```

The new profit function can now be defined.

In[14]:=
```
newprofit[p_] = p * demand[p] - newcost[demand[p]]
```
Out[14]=
```
-fixedcost - 31. (26000. - 400. p) + (26000. - 400. p) p
```

The derivative of the new profit function must be zero at a critical point.

In[15]:=
```
newmaxprofitcondition = D[newprofit[p], p] == 0
```
Out[15]=
```
38400. - 800. p == 0
```

Solving the equation obtained gives us the new profit maximizing price.

In[16]:=
```
newmaxprofitsoln = Solve[newmaxprofitcondition, p][[1]]
```
Out[16]=
```
{p → 48.}
```

In[17]:=
```
newmaxprofitprice = p /. newmaxprofitsoln
```
Out[17]=
```
48.
```

The approximate pass-through rate is then the ratio of change in optimal price over the change in marginal cost.

In[18]:=
```
approxpassthroughrate =
  (newmaxprofitprice - maxprofitprice) / (newmarginalcost - marginalcost)
```
Out[18]=
```
0.5
```

4. Instead of an increasing marginal cost, consider the effect of imposing a corporate profits tax of 10%. I.e., how do the optimal price and quantity change from question 2 if the firm maximizes the 90% of its profits that it gets to keep?

The optimal price and quantity don’t change, since the maximum profit condition remains the same. A factor of 0.9 in the equation will not change the solution to the equation.

In[19]:=
```
condition1 = D[profit[p], p] == 0
```
Out[19]=
```
38000. - 800. p == 0
```

In[20]:=
```
Solve[condition1, p][[1]]
```
Out[20]=
```
{p → 47.5}
```

In[21]:=
```
condition2 = D[0.9 * profit[p], p] == 0
```
Out[21]=
```
0.9 (38 000. - 800. p) == 0
```

In[22]:=
```
Solve[condition2, p][[1]]
```
Out[22]=
```
{p → 47.5}
```

5. Consider the effect of imposing a sales tax of 10%. Suppose that whatever price the firm decides on, the final price the consumer sees is 10% higher and that price is what determines the demand for the product, i.e., assume the same consumer demand function as in 1 is applied to 1.1 times the price the firm sets (and receives) for the product. What happens to optimal price and quantity if the firm selects a price that maximizes its profit given this change in consumer demand due to the tax?

The new demand function is given below.

In[23]:=
```
demandwithsalestax[p_] = demand[1.1 * p]
```
Out[23]=
```
26 000. - 440. p
```

We redefine the profit function.

In[24]:=
```
profit[p_] = p * demandwithsalestax[p] - cost[demandwithsalestax[p]]
```
Out[24]=
```
-fixedcost - 30. (26 000. - 440. p) + (26 000. - 440. p) p
```

We compute the optimal price again.

In[25]:=
```
maxprofitcondition = D[profit[p], p] == 0
```
Out[25]=
```
39 200. - 880. p == 0
```

In[26]:=
```
maxprofitsoln = Solve[maxprofitcondition, p][[1]]
```
Out[26]=
```
{p → 44.5455}
```

In[27]:=
```
maxprofitprice = p /. maxprofitsoln
```
Out[27]=
```
44.5455
```

We see that the optimal price has reduced.

With the sales tax, the new demand at the profit maximizing price is computed below.

In[28]:=
```
newoptimaldemand = demandwithsalestax[maxprofitprice]
```
Out[28]=
```
6400.
```

We see that the optimal demand is reduced with the sales tax.

6. This example has assumed that demand is linear. An alternative is to assume that demand is constant elasticity. The demand function in this case would be given by the formula given earlier. Repeat 1-5 with this demand form, calibrating the demand function to the same conditions, a demand of 10000 at $40 and 6000 at $50.

Define the new demand function.

```
In[29]:= cedemand[p_] = (scale) * p^elast
Out[29]=
```

p^{elast} scale

It’s easier to solve for the unknown constants if we take a logarithm of the entire equation. We define a new equation with new coefficients.

```
In[30]:= logdemand[p_] = PowerExpand[Log[cedemand[p]]]
Out[30]=
elast Log[p] + Log[scale]
```

We now impose the demand conditions.

```
In[31]:= demandconditions = {logdemand[40.] == Log[10000.], logdemand[50.] == Log[6000.]}
Out[31]=
{3.68888 elast + Log[scale] == 9.21034, 3.91202 elast + Log[scale] == 8.69951}
```

We solve for the coefficients.

```
In[32]:= coeffs = Solve[demandconditions, {elast, scale}][[1]]
Out[32]=
```

$\{elast \to -2.28922,\ scale \to 4.65023 \times 10^{7}\}$

We now have the exact form of the demand function.

```
In[33]:= cedemand[p_] = cedemand[p] /. coeffs
Out[33]=
```

$$\frac{4.65023 \times 10^{7}}{p^{2.28922}}$$

To compute the optimal demand and price, we devise a cost function with an unknown fixed cost, and the given marginal cost.

```
In[34]:= marginalcost = 30.
Out[34]=
30.
```

```
In[35]:= cost[q_] = fixedcost + marginalcost * q
Out[35]=
fixedcost + 30. q
```

The profit function can now be defined.

In[36]:=
```
profit[p_] = p * cedemand[p] - cost[cedemand[p]]
```
Out[36]=

$$-\text{fixedcost} - \frac{1.39507\times 10^{9}}{p^{2.28922}} + \frac{4.65023\times 10^{7}}{p^{1.28922}}$$

The derivative of the profit function must be zero at a critical point.

In[37]:=
```
maxprofitcondition = D[profit[p], p] == 0
```
Out[37]=

$$\frac{3.19363\times 10^{9}}{p^{3.28922}} - \frac{5.99519\times 10^{7}}{p^{2.28922}} == 0$$

Solving the equation obtained gives us the profit maximizing price.

In[38]:=
```
maxprofitsoln = FindRoot[maxprofitcondition, {p, 45.}][[1]]
```
Out[38]=

$p \to 53.2698$

In[39]:=
```
maxprofitprice = p /. maxprofitsoln
```
Out[39]=

53.2698

The demand at this profit maximizing price is computed below.

In[40]:=
```
demandatmaxprofit = cedemand[maxprofitprice]
```
Out[40]=

5190.06

To find the approximate pass-through rate, We devise a new cost function with an unknown fixed cost, and the new marginal cost.

In[41]:=
```
newmarginalcost = 31.
```
Out[41]=

31.

In[42]:=
```
newcost[q_] = fixedcost + newmarginalcost * q
```
Out[42]=

fixedcost + 31. q

The new profit function can now be defined.

In[43]:=
```
newprofit[p_] = p * cedemand[p] - newcost[cedemand[p]]
```
Out[43]=

$$-\text{fixedcost} - \frac{1.44157\times 10^{9}}{p^{2.28922}} + \frac{4.65023\times 10^{7}}{p^{1.28922}}$$

The derivative of the new profit function must be zero at a critical point.

In[44]:=
```
newmaxprofitcondition = D[newprofit[p], p] == 0
```
Out[44]=

$$\frac{3.30008\times 10^{9}}{p^{3.28922}} - \frac{5.99519\times 10^{7}}{p^{2.28922}} == 0$$

Solving the equation obtained gives us the new profit maximizing price.

In[45]:=
```
newmaxprofitsoln = FindRoot[newmaxprofitcondition, {p, 50.}][[1]]
```
Out[45]=

$p \to 55.0455$

In[46]:=
```
newmaxprofitprice = p /. newmaxprofitsoln
```
Out[46]=

55.0455

The approximate pass-through rate is then the ratio of change in optimal price over the change in marginal cost.

In[47]:=
```
approxpassthroughrate =
  (newmaxprofitprice - maxprofitprice) / (newmarginalcost - marginalcost)
```
Out[47]=

1.77566

The imposition of a corporate tax does not change the optimal price and quantity, since the equation for optimal demand price does not change.

Next we explore the effect of sales tax on the new demand model.

The new demand function is given below.

In[48]:=
```
demandwithsalestax[p_] = cedemand[1.1 * p]
```
Out[48]=

$$\frac{3.73867 \times 10^{7}}{p^{2.28922}}$$

We redefine the profit function.

In[49]:=
```
profit[p_] = p * demandwithsalestax[p] - cost[demandwithsalestax[p]]
```
Out[49]=

$$-\text{fixedcost} - \frac{1.1216 \times 10^{9}}{p^{2.28922}} + \frac{3.73867 \times 10^{7}}{p^{1.28922}}$$

We compute the optimal price again.

In[50]:=
```
maxprofitcondition = D[profit[p], p] == 0
```
Out[50]=

$$\frac{2.5676 \times 10^{9}}{p^{3.28922}} - \frac{4.81999 \times 10^{7}}{p^{2.28922}} == 0$$

In[51]:=
```
maxprofitsoln = FindRoot[maxprofitcondition, {p, 50.}][[1]]
```
Out[51]=

$p \to 53.2698$

In[52]:=
```
maxprofitprice = p /. maxprofitsoln
```
Out[52]=

53.2698

We see that the optimal price has not changed.

With the sales tax, the new demand at the profit maximizing price is computed below.

In[53]:=
```
newoptimaldemand = demandwithsalestax[maxprofitprice]
```
Out[53]=
```
4172.68
```

We see that the optimal demand is reduced with the sales tax.