Math 3660 - Spring 2022
Mathematical Models in Economics

Steven Tschantz

1/10/22

Profit maximizing firms

Assignment 2

Instructions

Save a copy of this notebook, complete the exercises, save and submit on Brightspace your final version by start of class Tues. Feb. 1.

Nikhil Jayswal

Edit the line above that reads Enter your name here, inserting instead your name. You may also want to save to a file name that includes your name. By adding your name into the file and file name, you reduce the chance I will mix-up your submission with someone else’s.

Exercises

Clear all of the symbols we will be using.

```
In[1]:= Clear[p1, p2, q1, q2, a11, a12, a21, a22, b1, b2, mc1, mc2, profit1, profit2];
```

As above, suppose the demands for products are 1 and 2 are linear functions given by

```
In[2]:= q1[p1_, p2_] := a11 * p1 + a12 * p2 + b1;
        q2[p1_, p2_] := a21 * p1 + a22 * p2 + b2;
```

1. At times x, y, and z, we observe the prices and demand for products 1 and 2, obtaining the following (hypothetical) data. Determine the coefficients a11,

a12, a21, a22, b1, and b2 of the linear demand functions that match this data. (Set up a system of equations, given as a list, and solve for the coefficients.)

In[4]:=
```
p1x = 50.; p2x = 40.; q1x = 100.; q2x = 130.;
p1y = 55.; p2y = 40.; q1y = 80.; q2y = 140.;
p1z = 50.; p2z = 45.; q1z = 110.; q2z = 100.;
TableForm[{{p1x, p2x, q1x, q2x}, {p1y, p2y, q1y, q2y}, {p1z, p2z, q1z, q2z}}, TableHeadings →
  {{"time x", "time y", "time z"}, {"price 1", "price 2", "demand 1", "demand 2"}}]
```

Out[7]//TableForm=

	price 1	price 2	demand 1	demand 2
time x	50.	40.	100.	130.
time y	55.	40.	80.	140.
time z	50.	45.	110.	100.

In[8]:=
```
allequations = {q1[p1x, p2x] == q1x, q1[p1y, p2y] == q1y, q1[p1z, p2z] == q1z,
  q2[p1x, p2x] == q2x, q2[p1y, p2y] == q2y, q2[p1z, p2z] == q2z}
```

Out[8]=
```
{50. a11 + 40. a12 + b1 == 100., 55. a11 + 40. a12 + b1 == 80., 50. a11 + 45. a12 + b1 == 110.,
 50. a21 + 40. a22 + b2 == 130., 55. a21 + 40. a22 + b2 == 140., 50. a21 + 45. a22 + b2 == 100.}
```

In[9]:=
```
coefficients = Solve[allequations, {a11, a12, a21, a22, b1, b2}][[1]]
```

Out[9]=
```
{a11 → -4., a12 → 2., a21 → 2., a22 → -6., b1 → 220., b2 → 270.}
```

Thus, the desired coefficients are

In[10]:=
```
{a11, a12, a21, a22, b1, b2} /. coefficients
```

Out[10]=
```
{-4., 2., 2., -6., 220., 270.}
```

2. Determine the own elasticities and cross elasticities for these products at time x. (The *Mathematica* function D[formula,variable] gives the partial derivative of the formula with respect to the variable.)

In[11]:=
```
elasticitymatrix[p1_, p2_] = {
  {D[q1[p1, p2], p1] * p1 / q1[p1, p2], D[q1[p1, p2], p2] * p2 / q1[p1, p2]},
  {D[q2[p1, p2], p1] * p1 / q2[p1, p2], D[q2[p1, p2], p2] * p2 / q2[p1, p2]}}
```

Out[11]=

$$\left\{\left\{\frac{\text{a11 p1}}{\text{b1} + \text{a11 p1} + \text{a12 p2}}, \frac{\text{a12 p2}}{\text{b1} + \text{a11 p1} + \text{a12 p2}}\right\}, \left\{\frac{\text{a21 p1}}{\text{b2} + \text{a21 p1} + \text{a22 p2}}, \frac{\text{a22 p2}}{\text{b2} + \text{a21 p1} + \text{a22 p2}}\right\}\right\}$$

We substitute the prices at time x into the above elasticity matrix.

In[12]:=
```
elasticities = elasticitymatrix[p1x, p2x] /. coefficients
```

Out[12]=
```
{{-2., 0.8}, {0.769231, -1.84615}}
```

In[13]:=
```
MatrixForm[elasticities]
```

Out[13]//MatrixForm=

$$\begin{pmatrix} -2. & 0.8 \\ 0.769231 & -1.84615 \end{pmatrix}$$

The above elasticity matrix shows the own and cross elasticities for the products at time x.

3. Suppose constant marginal costs for the products are as given below and take fixed costs to be 0. Define functions profit1 and profit2 for the profits on products 1 and 2 (profit1[p1_,p2_]:=... and profit2[p1_,p2_]:=...).

In[14]:=
```
mc1 = 25; mc2 = 15;
```

The profit functions are relatively straightforward, with profit being equal to the difference between total revenue and total costs.

In[15]:=
```
profit1[p1_, p2_] := p1 * q1[p1, p2] - mc1 * q1[p1, p2]
```

In[16]:=
```
profit2[p1_, p2_] := p2 * q2[p1, p2] - mc2 * q2[p1, p2]
```

4. If firm 1 sets it price at 50. what is the optimal price for firm 2? If firm 2 sets its price to this amount what is the optimal price for firm 1? And then if firm 1 takes this price what would firm 2 do, and then what would firm 1 respond?

Suppose that firm 1 sets its price at 50. This is represented by the variable ‘p1a’ below.

In[17]:=
```
p1a = 50.
```

Out[17]=
```
50.
```

The profit function of firm 2 when evaluated with p1 = 50 OR p1 = p1a is now a function of the price p2 only.

In[18]:=
```
profit2[p1a, p2] /. coefficients
```

Out[18]=
```
-15 (370. - 6. p2) + (370. - 6. p2) p2
```

In[19]:=
```
Plot[profit2[p1a, p2] /. coefficients, {p2, 0, 100}]
```

Out[19]=



We differentiate the function to find the price for which maxima occurs as shown in the plot above.

In[20]:=
```
p2a = Solve[D[profit2[p1a, p2] /. coefficients, p2] == 0, p2][[1]]
```

Out[20]=

```
{p2 → 38.3333}
```

In[21]:=
```
p2a = p2 /. p2a
```

Out[21]=

```
38.3333
```

Hence, we see that the optimum price for firm 2 is 38.33. We can substitute this for p2 to see what the profit function for firm 1 looks like now.

In[22]:=
```
profit1[p1, p2a] /. coefficients
```

Out[22]=

```
-25 (296.667 - 4. p1) + (296.667 - 4. p1) p1
```

In[23]:=
```
Plot[profit1[p1, p2a] /. coefficients, {p1, 0, 100}]
```

Out[23]=



We differentiate the function to find the price for which maxima occurs as shown in the plot above.

In[24]:=
```
p1b = Solve[D[profit1[p1, p2a] /. coefficients, p1] == 0, p1][[1]]
```
Out[24]=
```
{p1 → 49.5833}
```

In[25]:=
```
p1b = p1 /. p1b
```
Out[25]=
```
49.5833
```

Hence, we see that the optimum price for firm 1 is now 49.58. We can substitute this for p1 to see what the profit function for firm 2 looks like now.

In[26]:=
```
profit2[p1b, p2] /. coefficients
```
Out[26]=
```
-15 (369.167 - 6. p2) + (369.167 - 6. p2) p2
```

In[27]:=
```
Plot[profit2[p1b, p2] /. coefficients, {p2, 0, 100}]
```
Out[27]=



We differentiate the function to find the price for which maxima occurs as shown in the plot above.

In[28]:=
```
p2b = Solve[D[profit2[p1b, p2] /. coefficients, p2] == 0, p2][[1]]
```
Out[28]=
```
{p2 → 38.2639}
```

In[29]:=
```
p2b = p2 /. p2b
```
Out[29]=
```
38.2639
```

Hence, we see that the optimum price for firm 2 is now 38.26. We can substitute this for p2 to see what the profit function for firm 1 looks like now.

In[30]:=
```
profit1[p1, p2b] /. coefficients
```
Out[30]=
```
-25 (296.528 - 4. p1) + (296.528 - 4. p1) p1
```

In[31]:=
```
Plot[profit1[p1, p2b] /. coefficients, {p1, 0, 100}]
```
Out[31]=



We differentiate the function to find the price for which maxima occurs as shown in the plot above.

In[32]:=
```
p1c = Solve[D[profit1[p1, p2b] /. coefficients, p1] == 0, p1][[1]]
```
Out[32]=
```
{p1 → 49.566}
```

In[33]:=
```
p1c = p1 /. p1c
```
Out[33]=
```
49.566
```

Hence, we see that the optimum price for firm 1 is now 49.56.

5. Find the Nash equilibrium prices, i.e., the prices p1e and p2e such that profit1[p1,p2e] is a maximum at p1=p1e and profit2[p1e,p2] is a maximum at p2=p2e, so that firm 1 is doing the best it can given what firm 2 is doing, and firm 2 is doing the best it can given what firm 1 is doing.

To find the Nash equilibrium prices, we need to solve simultaneous equations.

In[34]:=
```
NashEquilibriumEquations = {D[profit1[p1, p2], p1] == 0, D[profit2[p1, p2], p2] == 0}
```
Out[34]=
```
{-25 a11 + b1 + 2 a11 p1 + a12 p2 == 0, -15 a22 + b2 + a21 p1 + 2 a22 p2 == 0}
```

In[35]:=
```
NashEquilibriumEquations = NashEquilibriumEquations /. coefficients
```
Out[35]=
```
{320. - 8. p1 + 2. p2 == 0, 360. + 2. p1 - 12. p2 == 0}
```

In[36]:=
```
NashEqulibirumPrices = Solve[NashEquilibriumEquations, {p1, p2}][[1]]
```
Out[36]=
```
{p1 → 49.5652, p2 → 38.2609}
```

In[37]:=
```
{p1Nash, p2Nash} = {p1, p2} /. NashEqulibirumPrices
```
Out[37]=
```
{49.5652, 38.2609}
```

In[38]:=
```
profit1Nash = profit1[p1Nash, p2Nash] /. coefficients
```
Out[38]=
```
2413.8
```

In[39]:=
```
profit2Nash = profit2[p1Nash, p2Nash] /. coefficients
```
Out[39]=
```
3246.41
```

Thus the prices for which Nash equilibrium is achieved is 49.56 for Firm 1 and 38.26 for Firm 2. At these prices, they have profits of 2413.8 and 3246.41 respectively.

6. If the firms merge, they will only be concerned about the total profit on the two products. Find the prices the merged firm will charge to maximize total profit, i.e., the prices p1m and p2m such that profit1[p1,p2]+profit2[p1,p2] is a maximum at p1=p1m and p2=p2m.

The total profit function is given below.

In[40]:=
```
totalprofit[p1_, p2_] := profit1[p1, p2] + profit2[p1, p2]
```

To maximize the total profit, we differentiate the total profit function w.r.t p1 and p2 and set the partial derivatives to zero.

In[41]:=
```
equation1 = D[totalprofit[p1, p2], p1] == 0
```
Out[41]=
```
-25 a11 - 15 a21 + b1 + 2 a11 p1 + a12 p2 + a21 p2 == 0
```

In[42]:=
```
equation2 = D[totalprofit[p1, p2], p2] == 0
```
Out[42]=
```
-25 a12 - 15 a22 + b2 + a12 p1 + a21 p1 + 2 a22 p2 == 0
```

In[43]:=
```
equations = {equation1, equation2} /. coefficients
```
Out[43]=
```
{290. - 8. p1 + 4. p2 == 0, 310. + 4. p1 - 12. p2 == 0}
```

Solving these equations will give us the prices that maximize the total profit.

In[44]:=
```
optimalPrices = Solve[equations, {p1, p2}][[1]]
```
Out[44]=
```
{p1 → 59., p2 → 45.5}
```

In[45]:=
```
{p1opt, p2opt} = {p1, p2} /. optimalPrices
```
Out[45]=
```
{59., 45.5}
```

In[46]:=
```
optimalProfit = totalprofit[p1opt, p2opt] /. coefficients
```
Out[46]=
```
6057.5
```

Thus, the total profit is maximized when the price for product of Firm 1 is 59, and the price for product of Firm 2 is 45.5. The optimized total profit is 6057.5.